Rules for integrands of the form $(a + b x^n)^p \operatorname{Sinh}[c + d x]$

1: $\int (a + b x^n)^p \operatorname{Sinh}[c + d x] \, dx$ when $p \in \mathbb{Z}^+$

Derivation: Algebraic expansion

Rule: If $p \in \mathbb{Z}^+$, then

$$\int (a + b x^n)^p \operatorname{Sinh}[c + d x] \, dx \longrightarrow \int \operatorname{Sinh}[c + d x] \operatorname{ExpandIntegrand}\left[(a + b x^n)^p, x\right] dx$$

```
Int[(a_+b_.*x_^n_)^p_.*Sinh[c_.+d_.*x_],x_Symbol] :=
  Int[ExpandIntegrand[Sinh[c+d*x],(a+b*x^n)^p,x],x] /;
FreeQ[{a,b,c,d,n},x] && IGtQ[p,0]
```

```
Int[(a_+b_.*x_^n_)^p_.*Cosh[c_.+d_.*x_],x_Symbol] :=
  Int[ExpandIntegrand[Cosh[c+d*x],(a+b*x^n)^p,x],x] /;
FreeQ[{a,b,c,d,n},x] && IGtQ[p,0]
```

2. $\int (a + b x^n)^p \operatorname{Sinh}[c + d x] \, dx$ when $p \in \mathbb{Z}^- \land n \in \mathbb{Z}$

1. $\int (a + b x^n)^p \operatorname{Sinh}[c + d x] \, dx$ when $p \in \mathbb{Z}^- \land n \in \mathbb{Z}^+$

1: $\int (a + b x^n)^p \operatorname{Sinh}[c + d x] \, dx$ when $p \in \mathbb{Z}^- \land n \in \mathbb{Z}^+ \land p < -1 \land n > 2$

Derivation: Integration by parts

Basis: $\partial_x \frac{(a+b x^n)^{p+1}}{b n (p+1)} == x^{n-1} (a + b x^n)^p$

Basis: $\partial_x \left(x^{-n+1} \operatorname{Sinh}[c + d x]\right) == -(n-1) x^{-n} \operatorname{Sinh}[c + d x] + d x^{-n+1} \operatorname{Cosh}[c + d x]$

Rule: If $p \in \mathbb{Z}^- \land n \in \mathbb{Z}^+ \land p < -1 \land n > 2$, then

$$\int (a + b x^n)^p \operatorname{Sinh}[c + d x] \, dx \longrightarrow$$

$$\frac{x^{-n+1}\left(a+b\,x^n\right)^{p+1}\operatorname{Sinh}[c+d\,x]}{b\,n\,(p+1)}-\frac{-n+1}{b\,n\,(p+1)}\int x^{-n}\left(a+b\,x^n\right)^{p+1}\operatorname{Sinh}[c+d\,x]\,dx-\frac{d}{b\,n\,(p+1)}\int x^{-n+1}\left(a+b\,x^n\right)^{p+1}\operatorname{Cosh}[c+d\,x]\,dx$$

Program code:

```
Int[(a_+b_.*x_^n_)^p_*Sinh[c_.+d_.*x_],x_Symbol] :=
  x^(-n+1)*(a+b*x^n)^(p+1)*Sinh[c+d*x]/(b*n*(p+1)) -
  (-n+1)/(b*n*(p+1))*Int[x^(-n)*(a+b*x^n)^(p+1)*Sinh[c+d*x],x] -
  d/(b*n*(p+1))*Int[x^(-n+1)*(a+b*x^n)^(p+1)*Cosh[c+d*x],x] /;
FreeQ[{a,b,c,d},x] && IntegerQ[p] && IGtQ[n,0] && LtQ[p,-1] && GtQ[n,2]
```

```
Int[(a_+b_.*x_^n_)^p_*Cosh[c_.+d_.*x_],x_Symbol] :=
  x^(-n+1)*(a+b*x^n)^(p+1)*Cosh[c+d*x]/(b*n*(p+1)) -
  (-n+1)/(b*n*(p+1))*Int[x^(-n)*(a+b*x^n)^(p+1)*Cosh[c+d*x],x] -
  d/(b*n*(p+1))*Int[x^(-n+1)*(a+b*x^n)^(p+1)*Sinh[c+d*x],x] /;
FreeQ[{a,b,c,d},x] && IntegerQ[p] && IGtQ[n,0] && LtQ[p,-1] && GtQ[n,2]
```

2: $\int\left(a+b\,x^n\right)^p\operatorname{Sinh}[c+d\,x]\,dx$ when $p\in\mathbb{Z}^-\land n\in\mathbb{Z}^+$

Derivation: Algebraic expansion

Rule: If $p\in\mathbb{Z}^-\land n\in\mathbb{Z}^+$, then

$$\int\left(a+b\,x^n\right)^p\operatorname{Sinh}[c+d\,x]\,dx\longrightarrow\int\operatorname{Sinh}[c+d\,x]\operatorname{ExpandIntegrand}\left[\left(a+b\,x^n\right)^p,\,x\right]dx$$

```
Int[(a_+b_.*x_^n_)^p_*Sinh[c_.+d_.*x_],x_Symbol] :=
  Int[ExpandIntegrand[Sinh[c+d*x],(a+b*x^n)^p,x],x] /;
FreeQ[{a,b,c,d},x] && ILtQ[p,0] && IGtQ[n,0] && (EqQ[n,2] || EqQ[p,-1])
```

```
Int[(a_+b_.*x_^n_)^p_*Cosh[c_.+d_.*x_],x_Symbol] :=
  Int[ExpandIntegrand[Cosh[c+d*x],(a+b*x^n)^p,x],x] /;
FreeQ[{a,b,c,d},x] && ILtQ[p,0] && IGtQ[n,0] && (EqQ[n,2] || EqQ[p,-1])
```

2: $\int (a+b\,x^n)^p\,\text{Sinh}[c+d\,x]\,dx$ when $p\in\mathbb{Z}^-\land n\in\mathbb{Z}^-$

Derivation: Algebraic simplification

Basis: If $p\in\mathbb{Z}$, then $(a+b\,x^n)^p == x^{n\,p}\,(b+a\,x^{-n})^p$

Rule: If $p\in\mathbb{Z}^-\land n\in\mathbb{Z}^-$, then

$$\int (a+b\,x^n)^p\,\text{Sinh}[c+d\,x]\,dx \longrightarrow \int x^{n\,p}\,(b+a\,x^{-n})^p\,\text{Sinh}[c+d\,x]\,dx$$

```
Int[(a_+b_.*x_^n_)^p_*Sinh[c_.+d_.*x_],x_Symbol] :=
  Int[x^(n*p)*(b+a*x^(-n))^p*Sinh[c+d*x],x] /;
FreeQ[{a,b,c,d},x] && ILtQ[p,0] && ILtQ[n,0]
```

```
Int[(a_+b_.*x_^n_)^p_*Cosh[c_.+d_.*x_],x_Symbol] :=
  Int[x^(n*p)*(b+a*x^(-n))^p*Cosh[c+d*x],x] /;
FreeQ[{a,b,c,d},x] && ILtQ[p,0] && ILtQ[n,0]
```

X: $\int (a+b\,x^n)^p\,\text{Sinh}[c+d\,x]\,dx$

Rule:

$$\int (a+b\,x^n)^p\,\text{Sinh}[c+d\,x]\,dx \longrightarrow \int (a+b\,x^n)^p\,\text{Sinh}[c+d\,x]\,dx$$

Program code:

```
Int[(a_+b_.*x_^n_)^p_*Sinh[c_.+d_.*x_],x_Symbol] :=
  Unintegrable[(a+b*x^n)^p*Sinh[c+d*x],x] /;
FreeQ[{a,b,c,d,n,p},x]
```

```
Int[(a_+b_.*x_^n_)^p_*Cosh[c_.+d_.*x_],x_Symbol] :=
  Unintegrable[(a+b*x^n)^p*Cosh[c+d*x],x] /;
FreeQ[{a,b,c,d,n,p},x]
```

Rules for integrands of the form $(e\,x)^m\,(a+b\,x^n)^p\,\text{Sinh}[c+d\,x]$

1: $\int (e\,x)^m\,(a+b\,x^n)^p\,\text{Sinh}[c+d\,x]\,dx$ when $p\in\mathbb{Z}^+$

Derivation: Algebraic expansion

Rule: If $p\in\mathbb{Z}^+$, then

$$\int (e\,x)^m\,(a+b\,x^n)^p\,\text{Sinh}[c+d\,x]\,dx\;\longrightarrow\;\int \text{Sinh}[c+d\,x]\;\text{ExpandIntegrand}\big[(e\,x)^m\,(a+b\,x^n)^p,\,x\big]\,dx$$

```
Int[(e_.*x_)^m_.*(a_+b_.*x_^n_)^p_.*Sinh[c_.+d_.*x_],x_Symbol] :=
  Int[ExpandIntegrand[Sinh[c+d*x],(e*x)^m*(a+b*x^n)^p,x],x] /;
FreeQ[{a,b,c,d,e,m,n},x] && IGtQ[p,0]
```

```
Int[(e_.*x_)^m_.*(a_+b_.*x_^n_)^p_.*Cosh[c_.+d_.*x_],x_Symbol] :=
  Int[ExpandIntegrand[Cosh[c+d*x],(e*x)^m*(a+b*x^n)^p,x],x] /;
FreeQ[{a,b,c,d,e,m,n},x] && IGtQ[p,0]
```

2: $\int (e\,x)^m \left(a+b\,x^n\right)^p \text{Sinh}[c+d\,x]\,dx$ when $p\in\mathbb{Z}^-\land m==n-1\land p<-1\land (n\in\mathbb{Z}\lor e>0)$

Derivation: Integration by parts

Basis: If $m==n-1\land (n\in\mathbb{Z}\lor e>0)$, then $\partial_x \frac{e^m (a+b\,x^n)^{p+1}}{b\,n\,(p+1)} == (e\,x)^m (a+b\,x^n)^p$

Rule: If $p\in\mathbb{Z}\land m==n-1\land p<-1\land (n\in\mathbb{Z}\lor e>0)$, then

$$\int (e\,x)^m \left(a+b\,x^n\right)^p \text{Sinh}[c+d\,x]\,dx \longrightarrow \frac{e^m \left(a+b\,x^n\right)^{p+1} \text{Sinh}[c+d\,x]}{b\,n\,(p+1)} - \frac{d\,e^m}{b\,n\,(p+1)} \int \left(a+b\,x^n\right)^{p+1} \text{Cosh}[c+d\,x]\,dx$$

```
Int[(e_.*x_)^m_.*(a_+b_.*x_^n_)^p_*Sinh[c_.+d_.*x_],x_Symbol] :=
  e^m*(a+b*x^n)^(p+1)*Sinh[c+d*x]/(b*n*(p+1)) -
  d*e^m/(b*n*(p+1))*Int[(a+b*x^n)^(p+1)*Cosh[c+d*x],x] /;
FreeQ[{a,b,c,d,e,m,n},x] && IntegerQ[p] && EqQ[m-n+1,0] && LtQ[p,-1] && (IntegerQ[n] || GtQ[e,0])
```

```
Int[(e_.*x_)^m_.*(a_+b_.*x_^n_)^p_*Cosh[c_.+d_.*x_],x_Symbol] :=
  e^m*(a+b*x^n)^(p+1)*Cosh[c+d*x]/(b*n*(p+1)) -
  d*e^m/(b*n*(p+1))*Int[(a+b*x^n)^(p+1)*Sinh[c+d*x],x] /;
FreeQ[{a,b,c,d,e,m,n},x] && IntegerQ[p] && EqQ[m-n+1,0] && LtQ[p,-1] && (IntegerQ[n] || GtQ[e,0])
```

3. $\int x^m \left(a+b x^n\right)^p \mathrm{Sinh}[c+d x]\, dx$ when $p \in \mathbb{Z}^- \wedge (m \mid n) \in \mathbb{Z}$

   1. $\int x^m \left(a+b x^n\right)^p \mathrm{Sinh}[c+d x]\, dx$ when $p \in \mathbb{Z}^- \wedge n \in \mathbb{Z}^+$

      1: $\int x^m \left(a+b x^n\right)^p \mathrm{Sinh}[c+d x]\, dx$ when $p+1 \in \mathbb{Z}^- \wedge n \in \mathbb{Z}^+ \wedge (m-n+1>0 \vee n>2)$

Derivation: Integration by parts

Basis: $\partial_x \frac{(a+b x^n)^{p+1}}{b n (p+1)} == x^{n-1} (a+b x^n)^p$

Basis: $\partial_x \left(x^{m-n+1} \mathrm{Sinh}[c+d x]\right) == (m-n+1)\, x^{m-n} \mathrm{Sinh}[c+d x] + d\, x^{m-n+1} \mathrm{Cosh}[c+d x]$

Rule: If $p+1 \in \mathbb{Z}^- \wedge n \in \mathbb{Z}^+ \wedge (m-n+1>0 \vee n>2)$, then

$$\int x^m \left(a+b x^n\right)^p \mathrm{Sinh}[c+d x]\, dx \longrightarrow$$

$$\frac{x^{m-n+1} \left(a+b x^n\right)^{p+1} \mathrm{Sinh}[c+d x]}{b n (p+1)} - \frac{m-n+1}{b n (p+1)} \int x^{m-n} \left(a+b x^n\right)^{p+1} \mathrm{Sinh}[c+d x]\, dx - \frac{d}{b n (p+1)} \int x^{m-n+1} \left(a+b x^n\right)^{p+1} \mathrm{Cosh}[c+d x]\, dx$$

Program code:

```
Int[x_^m_.*(a_+b_.*x_^n_)^p_*Sinh[c_.+d_.*x_],x_Symbol] :=
  x^(m-n+1)*(a+b*x^n)^(p+1)*Sinh[c+d*x]/(b*n*(p+1)) -
  (m-n+1)/(b*n*(p+1))*Int[x^(m-n)*(a+b*x^n)^(p+1)*Sinh[c+d*x],x] -
  d/(b*n*(p+1))*Int[x^(m-n+1)*(a+b*x^n)^(p+1)*Cosh[c+d*x],x] /;
FreeQ[{a,b,c,d},x] && ILtQ[p,-1] && IGtQ[n,0] && RationalQ[m] && (GtQ[m-n+1,0] || GtQ[n,2])
```

```
Int[x_^m_.*(a_+b_.*x_^n_)^p_*Cosh[c_.+d_.*x_],x_Symbol] :=
  x^(m-n+1)*(a+b*x^n)^(p+1)*Cosh[c+d*x]/(b*n*(p+1)) -
  (m-n+1)/(b*n*(p+1))*Int[x^(m-n)*(a+b*x^n)^(p+1)*Cosh[c+d*x],x] -
  d/(b*n*(p+1))*Int[x^(m-n+1)*(a+b*x^n)^(p+1)*Sinh[c+d*x],x] /;
FreeQ[{a,b,c,d},x] && ILtQ[p,-1] && IGtQ[n,0] && RationalQ[m] && (GtQ[m-n+1,0] || GtQ[n,2])
```

2: $\int x^m (a + b x^n)^p \operatorname{Sinh}[c + d x] \, dx$ when $p \in \mathbb{Z}^- \land n \in \mathbb{Z}^+$

Derivation: Algebraic expansion

–

Rule: If $p \in \mathbb{Z}^- \land n \in \mathbb{Z}^+$, then

$$\int x^m (a + b x^n)^p \operatorname{Sinh}[c + d x] \, dx \longrightarrow \int \operatorname{Sinh}[c + d x] \operatorname{ExpandIntegrand}[x^m (a + b x^n)^p, x] \, dx$$

```
Int[x_^m_.*(a_+b_.*x_^n_)^p_*Sinh[c_.+d_.*x_],x_Symbol] :=
  Int[ExpandIntegrand[Sinh[c+d*x],x^m*(a+b*x^n)^p,x],x] /;
FreeQ[{a,b,c,d},x] && ILtQ[p,0] && IntegerQ[m] && IGtQ[n,0] && (EqQ[n,2] || EqQ[p,-1])
```

```
Int[x_^m_.*(a_+b_.*x_^n_)^p_*Cosh[c_.+d_.*x_],x_Symbol] :=
  Int[ExpandIntegrand[Cosh[c+d*x],x^m*(a+b*x^n)^p,x],x] /;
FreeQ[{a,b,c,d},x] && ILtQ[p,0] && IntegerQ[m] && IGtQ[n,0] && (EqQ[n,2] || EqQ[p,-1])
```

2: $\int x^m (a + b x^n)^p \operatorname{Sinh}[c + d x] \, dx$ when $p \in \mathbb{Z}^- \land n \in \mathbb{Z}^-$

Derivation: Algebraic simplification

Basis: If $p \in \mathbb{Z}$, then $(a + b x^n)^p == x^{n p} (b + a x^{-n})^p$

–

Rule: If $p \in \mathbb{Z}^- \land n \in \mathbb{Z}^-$, then

$$\int x^m (a + b x^n)^p \operatorname{Sinh}[c + d x] \, dx \longrightarrow \int x^{m+n p} (b + a x^{-n})^p \operatorname{Sinh}[c + d x] \, dx$$

```
Int[x_^m_.*(a_+b_.*x_^n_)^p_*Sinh[c_.+d_.*x_],x_Symbol] :=
  Int[x^(m+n*p)*(b+a*x^(-n))^p*Sinh[c+d*x],x] /;
FreeQ[{a,b,c,d,m},x] && ILtQ[p,0] && ILtQ[n,0]
```

```
Int[x_^m_.*(a_+b_.*x_^n_)^p_*Cosh[c_.+d_.*x_],x_Symbol] :=
  Int[x^(m+n*p)*(b+a*x^(-n))^p*Cosh[c+d*x],x] /;
FreeQ[{a,b,c,d,m},x] && ILtQ[p,0] && ILtQ[n,0]
```

X: $\int (e\,x)^m\,(a+b\,x^n)^p\,\text{Sinh}[c+d\,x]\,dx$

Rule:

$$\int (e\,x)^m\,(a+b\,x^n)^p\,\text{Sinh}[c+d\,x]\,dx \longrightarrow \int (e\,x)^m\,(a+b\,x^n)^p\,\text{Sinh}[c+d\,x]\,dx$$

Program code:

```
Int[(e_.*x_)^m_.*(a_+b_.*x_^n_)^p_.*Sinh[c_.+d_.*x_],x_Symbol] :=
  Unintegrable[(e*x)^m*(a+b*x^n)^p*Sinh[c+d*x],x] /;
FreeQ[{a,b,c,d,e,m,n,p},x]
```

```
Int[(e_.*x_)^m_.*(a_+b_.*x_^n_)^p_.*Cosh[c_.+d_.*x_],x_Symbol] :=
  Unintegrable[(e*x)^m*(a+b*x^n)^p*Cosh[c+d*x],x] /;
FreeQ[{a,b,c,d,e,m,n,p},x]
```